
pkgname=gmp
pkgver=6.1.2
pkgrel=2
pkgdesc="A free library for arbitrary precision arithmetic"
arch=('x86_64')
url="https://gmplib.org/"
depends=('gcc-libs' 'sh')
license=('LGPL3' 'GPL')
options=('!libtool')
source=("https://ftp.gnu.org/gnu/gmp/gmp-${pkgver}.tar.xz")
md5sums=('f58fa8001d60c4c77595fbbb62b63c1d')

build() {
  cd ${pkgname}-${pkgver}
  
  ./configure --build=${CHOST} \
    --prefix=/usr \
    --enable-cxx \
    --enable-fat
  make
}

check() {
  cd ${pkgname}-${pkgver}
  
  make check
}

package() {
  cd ${pkgname}-${pkgver}
  
  make DESTDIR=${pkgdir} install
}
